Classroom Tips and Techniques:

Eigenvalue Problems for ODEs - Part 2

Robert J. Lopez

Emeritus Professor of Mathematics and Maple Fellow

Maplesoft

Initializations

> *restart*

> *with*(*VectorCalculus*) :
with(*PDEtools*) :

Introduction

In Part 1 of this series of articles on solving eigenvalue problems for ODEs, we discussed equations for which the general solution readily yielded eigenvalues and eigenfunctions without the need for detailed knowledge of any of the special functions of applied mathematics. In Part 2 of this series, we concentrate on eigenvalue problems for Bessel's equation whose solution requires some knowledge of Bessel functions of the first and second kinds.

Since Bessel's equation readily arises when separating variables in Laplace's equation in cylindrical coordinates, we have allowed our discusion to be colored by references to calculations arising from boundary value problems posed in a cylinder.

Steady-State Temperatures in a Cylinder

At steady state, the temperature u in a cylinder satisfies Laplace's equation $\nabla^2 u = 0$ and some conditions on the boundary of the cylinder. For example, if the cylinder is described in cylindrical coordinates by $0 \le r \le c$, $0 \le \theta \le 2\pi$, $0 \le z \le h$, we could impose the condition $u(r, \theta, h) = f(r)$ or $u(r, \theta, h) = F(r, \theta)$ on the top surface, $u(c, \theta, z) = 0$ or $u_r(c, \theta, z) = 0$ on the curved wall of the cylinder, and $u(r, \theta, 0) = 0$ or $u_z(r, \theta, 0) = 0$ on the bottom surface. (Subscripts denote partial derivatives.)

If the temperature on the curved wall of the cylinder is fixed at $u = 0$, we say that a homogeneous Dirichlet condition has been imposed. Alternatively, if the curved wall is insulated so the heat flux across this surface is zero, we say that a homogeneous Dirichlet condition has been imposed. The flux across the curved wall is the normal derivative given by $u_r = \frac{\partial u}{\partial r}$ evaluated at $r = 0$.

If the prescribed temperature on the top surface is $f(r)$, a function of r alone, the temperature in the cylinder will be axially symmetric so that $u(r, \theta, z) = u(r, z)$. Alternatively, if this prescribed temperature is $F(r, \theta)$, then the temperature in the cylinder will be axially asymmetric so that $u = u(r, \theta, z)$.

Axial Symmetry

Separation of Variables

Steady-state temperatures in a cylinder obey Laplace's equation. Under the assumption of axial symmetry, Laplace's equation in cylindrical coordinates becomes

> $PDE := expand(Laplacian(u(r, z), cylindrical[r, \theta, z])) = 0$

$$PDE := \frac{\frac{\partial}{\partial r} u(r, z)}{r} + \frac{\partial^2}{\partial r^2} u(r, z) + \frac{\partial^2}{\partial z^2} u(r, z) = 0$$

Assuming the separated solution

> $U := R(r)\, Z(z)$

$$U := R(r)\, Z(z)$$

leads to

> $q_1 := expand\left(\frac{PDE \Big|_{u(r, z) = U}}{U}\right)$

$$q_1 := \frac{\frac{d}{dr} R(r)}{R(r)\, r} + \frac{\frac{d^2}{dr^2} R(r)}{R(r)} + \frac{\frac{d^2}{dz^2} Z(z)}{Z(z)} = 0$$

and

> $temp := select(has, lhs(q_1), z):$
> $q_2 := q_1 - temp$

$$q_2 := \frac{\frac{d}{dr} R(r)}{R(r)\, r} + \frac{\frac{d^2}{dr^2} R(r)}{R(r)} = -\frac{\frac{d^2}{dz^2} Z(z)}{Z(z)}$$

as the variable-separated equation. Introducing the Bernoulli separation constant λ, we have the ordinary differential equation

> $q_3 := lhs(q_2) = \lambda$

$$q_3 := \frac{\frac{d}{dr} R(r)}{R(r)\, r} + \frac{\frac{d^2}{dr^2} R(r)}{R(r)} = \lambda$$

for the radial component $R(r)$. Manipulating this to the form

> $q_4 := expand(q_3 R(r))$

$$q_4 := \frac{\frac{d}{dr} R(r)}{r} + \frac{d^2}{dr^2} R(r) = R(r)\, \lambda$$

and then

> $q_5 := lhs(q_4) - rhs(q_4) = 0$

$$q_5 := \frac{\frac{d}{dr} R(r)}{r} + \frac{d^2}{dr^2} R(r) - R(r)\, \lambda = 0$$

we obtain the solution as

> $dsolve(q_5, R(r))$

$$R(r) = \_C1 \operatorname{BesselJ}(0, \sqrt{-\lambda}\, r) + \_C2 \operatorname{BesselY}(0, \sqrt{-\lambda}\, r)$$

The "self-adjoint" form of the equation for the radial component is

$$(rR')' + R' + \lambda\, r\, R = 0$$

The radical in the solution suggests the substitution

> $q_6 := q_5 \Big|_{\lambda = -\mu^2}$

$$q_6 := \frac{\frac{d}{dr} R(r)}{r} + \frac{d^2}{dr^2} R(r) + R(r)\, \mu^2 = 0$$

in which case the solution is given by

> $dsolve(q_6, R(r))$

$$R(r) = \_C1 \operatorname{BesselJ}(0, \mu\, r) + \_C2 \operatorname{BesselY}(0, \mu\, r)$$

The solution is a linear combination of two Bessel functions, one of the "first kind" and one of the "second."

The radial equation can be transformed to Bessel's equation by the change of variables

> $q_7 := dchange\left(\left\{r = \frac{x}{\mu}, R(r) = y(x)\right\}, q_6, [y(x), x]\right)$

$$q_7 := \frac{\mu^2 \left(\frac{d}{dx} y(x)\right)}{x} + \mu^2 \left(\frac{d^2}{dx^2} y(x)\right) + y(x)\,\mu^2 = 0$$

Factoring μ^2 leads to Bessel's equation of order zero, that is, to

> $q_8 := expand\left(\frac{q_7}{\mu^2}\right)$

$$q_8 := \frac{\frac{d}{dx} y(x)}{x} + \frac{d^2}{dx^2} y(x) + y(x) = 0$$

The solution of this equation is

> $dsolve(q_8, y(x))$

$$y(x) = \_C1\ \mathrm{BesselJ}(0, x) + \_C2\ \mathrm{BesselY}(0, x)$$

so that $R(r) = y(\mu\, r)$. Figure 1 contains a graph of $J_0(x)$ (in black) and $Y_0(x)$ (in red).

> $plot([\mathrm{BesselJ}(0, x), \mathrm{BesselY}(0, x)], x = 0..5, -4..2, color = [black, red])$



Figure 1 In black, $J_0(x)$, the Bessel function of the first kind, and in red, $Y_0(x)$, the Bessel function of the second kind.

Consistent with the graph in Figure 1, we have

> $\lim_{x \to 0+} \text{BesselY}(0, x) = \lim_{x \to 0+} \text{BesselY}(0, x)$

$$\lim_{x \to 0+} \text{BesselY}(0, x) = -\infty$$

so that the y-axis in a vertical asymptote for $Y_0(x)$, and $Y_0(x)$ is unbounded in any interval that includes $x = 0$.

The function $J_0(x)$ is well approximated by

> $g := \cos\left(x - \frac{\pi}{4}\right) \sqrt{\frac{2}{\pi x}}$

$$g := \sin\left(x + \frac{1}{4}\pi\right) \sqrt{2} \sqrt{\frac{1}{\pi x}}$$

as we confirm in Figure 2, where $J_0(x)$ is graphed in black, and $f(x)$ is in red.

>

> $plot([\mathrm{BesselJ}(0, x), g], x = 0..31, -0.55..1, color = [black, red], thickness = [5, 2])$



Figure 2 In black, $J_0(x)$, and in red, $f(x) = \sqrt{\frac{2}{\pi x}} \cos\left(x - \frac{\pi}{4}\right)$

Ignoring Maple's automatically simplifications of $\cos\left(x - \frac{\pi}{4}\right)$ to $\sin\left(x + \frac{\pi}{4}\right)$ and $\sqrt{\frac{2}{\pi x}}$ to $\sqrt{2}\sqrt{\frac{1}{\pi x}}$, we see from

> $\lim_{x \to \infty} (\mathrm{BesselJ}(0, x) - g) = \lim_{x \to \infty} (\mathrm{BesselJ}(0, x) - g)$

$$\lim_{x \to \infty} \left(\mathrm{BesselJ}(0, x) - \sin\left(x + \frac{1}{4}\pi\right)\sqrt{2}\sqrt{\frac{1}{\pi x}}\right) = 0$$

that $J_0(x)$ is asymptotic to $g(x)$.

Homogeneous Dirichlet Condition

The solution of the singular Sturm-Liouville eigenvalue problem

$$(r R')' + R' + \mu^2 R = 0$$
$$R(c) = 0$$
$$R(r) \text{ bounded on } [0, c]$$

consists of the eigenvalues μ_k (the zeros of $R(c) = y(\mu c) = J_0(\mu c)$) and the eigenfunctions $R_k(r) = J_0(\mu_k r)$, $k = 1, \ldots$.

The numbers $c\,\mu_k$, $k = 1,\ldots, 10$, are represented symbolically by

> $q_9 := \text{BesselJZeros}(0, 1..10)$

$$q_9 := \text{BesselJZeros}(0, 1..10)$$

and in floating-point form by

> $evalf(q_9)$

2.404825558, 5.520078110, 8.653727913, 11.79153444, 14.93091771, 18.07106397, 21.21163663, 24.35247153, 27.49347913, 30.63460647

That Maple recognizes the eigenvalues as exact is shown by

> $seq(\text{BesselJ}(0, \text{BesselJZeros}(0, k)), k = 1..10)$

$$0, 0, 0, 0, 0, 0, 0, 0, 0, 0$$

If, for example $c = 2$, the eigenvalues μ_k would be

> $evalf\left(\frac{[q_9]}{2}\right)$

[1.202412779, 2.760039055, 4.326863956, 5.895767220, 7.465458855, 9.035531985, 10.60581832, 12.17623576, 13.74673956, 15.31730324]

Figure 3 shows the first three eigenfunctions for the case $c = 2$.

> $plot\left(\left[seq\left(\text{BesselJ}\left(0, \frac{\text{BesselJZeros}(0, k)\,r}{2}\right), k = 1..3\right)\right], r = 0..2, color = [black, red, green]\right)$



Figure 3 For $c = 2$, the first three eigenfunctions in

black, red, and green, respectively

The eigenfunctions are orthogonal with respect to the weight function $w(r) = r$, as can be seen from the integral

> $temp := \int_0^c r \, \text{BesselJ}(0, \mu_j r) \, \text{BesselJ}(0, \mu_k r) \, dr$:

$temp = value(temp)$

$$\int_0^c r \, \text{BesselJ}(0, \mu_j r) \, \text{BesselJ}(0, \mu_k r) \, dr = \frac{c \left(-\text{BesselJ}(0, \mu_j c) \, \text{BesselJ}(1, \mu_k c) \, \mu_k + \text{BesselJ}(1, \mu_j c) \, \mu_j \, \text{BesselJ}(0, \mu_k c) \right)}{\mu_j^2 - \mu_k^2} \quad \textbf{(4.2.1)}$$

which vanishes if μ_j and μ_k are distinct eigenvalues for the interval $[0, c]$. Making use of Maple's built-in representation of the zeros of $J_0(\mu\, c)$, we also have

> $eval\left(\textbf{(4.2.1)}, \left[\mu_j = \frac{\text{BesselJZeros}(0, n)}{c}, \mu_k = \frac{\text{BesselJZeros}(0, m)}{c}\right]\right)$

$$\int_0^c r \, \text{BesselJ}\left(0, \frac{\text{BesselJZeros}(0, n)\, r}{c}\right) \text{BesselJ}\left(0, \frac{\text{BesselJZeros}(0, m)\, r}{c}\right) dr = 0$$

Unfortunately, Maple does not yet detect that this result is not valid if $n = m$, in which case the integral evaluates to

> $\lim_{\mu_j \to \mu_k} rhs(\textbf{(4.2.1)})$

$$\frac{1}{2} c^2 \, \text{BesselJ}(1, \mu_k c)^2 + \frac{1}{2} c^2 \, \text{BesselJ}(0, \mu_k c)^2$$

Since $J_0(\mu_k c) = 0$, this result simplifies to $\frac{c^2}{2} J_1^2(\mu_k c)$. Thus, under suitable conditions, a function $f(r)$ can be represented by the Fourier-Bessel series $\sum_{k=1}^{\infty} A_k J_0(\mu_k r)$, where

$$A_k = \frac{2}{c^2 J_1^2(\mu_k c)} \int_0^c r\, f(r)\, J_0(\mu_k r)\, dr$$

and the eigenvalue μ_k is the kth zero of $J_0(\mu\, c)$.

Homogeneous Neumann Condition

The solution of the singular Sturm-Liouville eigenvalue problem

$$(r R')' + R' + \sigma^2 R = 0$$
$$R'(c) = 0$$
$$R(r) \text{ bounded on } [0, c]$$

consists of the eigenvalues σ_k (the zeros of $R'(c) = y'(\sigma c) = J_0'(\mu c)$) and the eigenfunctions $R_k(r) = J_0(\sigma_k r)$, $k = 0, 1, \ldots$.

The numbers $c\,\sigma_k$, $k = 0, 1, \ldots, 10$, are represented symbolically by

> $q_{10} := \mathrm{BesselJZeros}(1, 0\,..10)$

$$q_{10} := 0, \mathrm{BesselJZeros}(1, 1\,..10)$$

and in floating-point form by

> $evalf(q_{10})$

0., 3.831705970, 7.015586670, 10.17346814, 13.32369194, 16.47063005, 19.61585851, 22.76008438, 25.90367209, 29.04682854, 32.18967991

>

where we have used the relationship $J_0'(x) = -J_1(x)$. Note carefully that $J_0(0) = 1$, but $J_n(0) = 0$ for positive integers n.

Figure 4 shows the first three eigenfunctions for the case $c = 1$.

> $plot([seq(\mathrm{BesselJ}(0, \mathrm{BesselJZeros}(1, k)\,x), k = 0\,..2)], x = 0\,..1, color = [black, red, green])$



Figure 4 For $c = 1$, the first three eigenfunctions in

black, red, and green, respectively

The eigenfunctions are orthogonal with respect to the weight function $w(r) = r$, as can be seen from the integral

> $temp := \int_0^c r\,\mathrm{BesselJ}(0, \sigma_j r)\,\mathrm{BesselJ}(0, \sigma_k r)\,\mathrm{d}r :$
>
> $temp = value(temp)$

$$\int_0^c r\,\mathrm{BesselJ}(0, \sigma_j r)\,\mathrm{BesselJ}(0, \sigma_k r)\,\mathrm{d}r = \frac{c\left(\mathrm{BesselJ}(0, \sigma_j c)\,\mathrm{BesselJ}(1, \sigma_k c)\,\sigma_k - \mathrm{BesselJ}(1, \sigma_j c)\,\sigma_j\,\mathrm{BesselJ}(0, \sigma_k c)\right)}{-\sigma_j^2 + \sigma_k^2} \quad \textbf{(4.3.1)}$$

which vanishes if σ_j and σ_k are distinct eigenvalues for the interval $[0, c]$. Of course, Maple has used $-J_1(x)$ for $J'_0(x)$.

Making use of Maple's built-in representation of the zeros of $J'_0(\sigma\,c)$, we also have

> $eval\left(\textbf{(4.3.1)}, \left[\sigma_j = \frac{\mathrm{BesselJZeros}(1, n)}{c}, \sigma_k = \frac{\mathrm{BesselJZeros}(1, m)}{c}\right]\right)$

$$\int_0^c r\,\mathrm{BesselJ}\left(0, \frac{\mathrm{BesselJZeros}(1, n)\,r}{c}\right)\mathrm{BesselJ}\left(0, \frac{\mathrm{BesselJZeros}(1, m)\,r}{c}\right)\mathrm{d}r = 0$$

As seen earlier, Maple does not yet detect that this result is not valid if $n = m$, in which case the integral evaluates to

> $\lim_{\sigma_j \to \sigma_k} rhs(\textbf{(4.3.1)})$

$$\frac{1}{2}\,c^2\,\mathrm{BesselJ}(1, \sigma_k c)^2 + \frac{1}{2}\,c^2\,\mathrm{BesselJ}(0, \sigma_k c)^2$$

Since $-J_1(\sigma_k c) = J'_0(\sigma_k c) = 0$, this result simplifies to $\frac{c^2}{2} J_0^2(\sigma_k c)$. Thus, under suitable conditions, a function $f(r)$ can be represented by the Fourier-Bessel series $\sum_{k=0}^{\infty} A_k J_0(\sigma_k r)$, where

$$A_0 = \frac{2}{c^2} \int_0^c r\,f(r)\,dr$$

$$A_k = \frac{2}{c^2 J_0^2(\sigma_k c)} \int_0^c r\,f(r)\,J_0(\sigma_k r)\,dr,\ k = 1, \ldots$$

and the eigenvalue σ_k is the kth zero of $J_1(\sigma c)$.

Axial Asymmetry

Separation of Variables

If the temperature prescribed on top of the cylinder is given by $u(r, \theta, h) = F(r, \theta)$, then $u(r, \theta, z)$ obeys Laplace's equation in the form

> $q_{11} := expand(Laplacian(u(r, \theta, z), cylindrical[r, \theta, z])) = 0$

$$q_{11} := \frac{\frac{\partial}{\partial r} u(r, \theta, z)}{r} + \frac{\partial^2}{\partial r^2} u(r, \theta, z) + \frac{\frac{\partial^2}{\partial \theta^2} u(r, \theta, z)}{r^2} + \frac{\partial^2}{\partial z^2} u(r, \theta, z) = 0$$

The separation assumption

> $U3 := R(r)\, \Theta(\theta)\, Z(z)$

$$U3 := R(r)\, \Theta(\theta)\, Z(z)$$

leads to

> $q_{12} := expand\left(\frac{q_{11}\Big|_{u(r, \theta, z) = U3}}{U3}\right)$

$$q_{12} := \frac{\frac{d}{dr} R(r)}{R(r)\, r} + \frac{\frac{d^2}{dr^2} R(r)}{R(r)} + \frac{\frac{d^2}{d\theta^2} \Theta(\theta)}{\Theta(\theta)\, r^2} + \frac{\frac{d^2}{dz^2} Z(z)}{Z(z)} = 0$$

Moving the term in z to the right-hand side leaves us with the equation

> $temp := select(has, lhs(q_{12}), z) :$
> $q_{13} := q_{12} - temp$

$$q_{13} := \frac{\frac{d}{dr} R(r)}{R(r)\, r} + \frac{\frac{d^2}{dr^2} R(r)}{R(r)} + \frac{\frac{d^2}{d\theta^2} \Theta(\theta)}{\Theta(\theta)\, r^2} = -\frac{\frac{d^2}{dz^2} Z(z)}{Z(z)}$$

Introducing the Bernoulli separation constant λ then gives

> $q_{14} := lhs(q_{13}) = -\lambda$
> $q_{15} := rhs(q_{13}) = -\lambda$

$$q_{14} := \frac{\frac{d}{dr} R(r)}{R(r)\, r} + \frac{\frac{d^2}{dr^2} R(r)}{R(r)} + \frac{\frac{d^2}{d\theta^2} \Theta(\theta)}{\Theta(\theta)\, r^2} = -\lambda$$

$$q_{15} := -\frac{\frac{d^2}{dz^2} Z(z)}{Z(z)} = -\lambda$$

The first of these equations is of immediate interest. We first multiply through by r^2 so that the term containing $\Theta(\theta)$ contains only the independent variable θ. This gives the equation

> $q_{16} := expand\left(r^2\, q_{14}\right)$

$$q_{16} := \frac{r\left(\frac{d}{dr} R(r)\right)}{R(r)} + \frac{r^2\left(\frac{d^2}{dr^2} R(r)\right)}{R(r)} + \frac{\frac{d^2}{d\theta^2} \Theta(\theta)}{\Theta(\theta)} = -r^2\, \lambda$$

Next, we would like to move the term $-r^2\,\lambda$ to the left, and the term containing θ to the right. The first transformation is given by

> $q_{17} := lhs(q_{16}) - rhs(q_{16}) = 0$

$$q_{17} := \frac{r\left(\frac{d}{dr} R(r)\right)}{R(r)} + \frac{r^2\left(\frac{d^2}{dr^2} R(r)\right)}{R(r)} + \frac{\frac{d^2}{d\theta^2} \Theta(\theta)}{\Theta(\theta)} + r^2\, \lambda = 0$$

and the second, by

> $temp := select\left(has, lhs(q_{17}), \theta\right):$
> $q_{18} := q_{17} - temp$

$$q_{18} := \frac{r\left(\frac{d}{dr} R(r)\right)}{R(r)} + \frac{r^2\left(\frac{d^2}{dr^2} R(r)\right)}{R(r)} + r^2\, \lambda = -\frac{\frac{d^2}{d\theta^2} \Theta(\theta)}{\Theta(\theta)}$$

Introducing a new separation constant ν, we have the two ordinary differential equations

> $q_{19} := lhs(q_{18}) = \nu$
> $q_{20} := rhs(q_{18}) = \nu$

$$q_{19} := \frac{r\left(\frac{d}{dr} R(r)\right)}{R(r)} + \frac{r^2\left(\frac{d^2}{dr^2} R(r)\right)}{R(r)} + r^2\, \lambda = \nu$$

$$q_{20} := -\frac{\frac{d^2}{d\theta^2} \Theta(\theta)}{\Theta(\theta)} = \nu$$

The second equation can be put into the form

> $ThetaEquation := expand(-(q_{20} - \nu)\,\Theta(\theta))$

$$ThetaEquation := \frac{d^2}{d\theta^2}\Theta(\theta) + \Theta(\theta)\,\nu = 0$$

and has general solution

> $T := unapply(rhs(dsolve(ThetaEquation, \Theta(\theta))), \theta):$
> $\Theta(\theta) = T(\theta)$

$$\Theta(\theta) = \_C1 \sin(\sqrt{\nu}\,\theta) + \_C2 \cos(\sqrt{\nu}\,\theta)$$

Continuity of $u(r, \theta, z)$ implies the periodic boundary conditions

$$\Theta(-\pi) = \Theta(\pi)$$
$$\Theta'(-\pi) = \Theta'(\pi)$$

Imposing these conditions leads to

> $T(-\pi) - T(\pi) = 0$
> $D(T)(-\pi) - D(T)(\pi) = 0$

$$-2\,\_C1 \sin(\sqrt{\nu}\,\pi) = 0$$
$$2\,\_C2 \sin(\sqrt{\nu}\,\pi)\sqrt{\nu} = 0$$

from which we see that $\nu = n^2$, $n = 0, 1, \ldots.$ Notice that for $n = 0$, the eigenspace has dimension 1 and basis $\{1\}$, but for $n \geq 1$ the eigenspaces have dimension 2 and bases $\{\cos(n\,\theta), \sin(n\,\theta)\}$.

The equation for the radial component $R(r)$ now takes the form

> $q_{21} := expand\left(\frac{q_{19}\,R(r)}{r}\right)\Big|_{\nu = n^2}$

$$q_{21} := \frac{d}{dr}R(r) + r\left(\frac{d^2}{dr^2}R(r)\right) + r\,R(r)\,\lambda = \frac{R(r)\,n^2}{r}$$

After moving all terms to the left, we obtain the self-adjoint form as

> $q_{22} := collect(lhs(q_{21}) - rhs(q_{21}), R(r)) = 0$

$$q_{22} := \left(r\,\lambda - \frac{n^2}{r}\right)R(r) + \frac{d}{dr}R(r) + r\left(\frac{d^2}{dr^2}R(r)\right) = 0$$

The solution of this equation is given by

> $dsolve(q_{22}, R(r))$

$$R(r) = \_C1\,\mathrm{BesselJ}(n, \sqrt{\lambda}\,r) + \_C2\,\mathrm{BesselY}(n, \sqrt{\lambda}\,r)$$

suggesting first, the substitution $\lambda = \mu^2$ so that the equation for the radial component becomes

$> q_{23} := q_{22}\Big|_{\lambda=\mu^2}$

$$q_{23} := \left(r\,\mu^2 - \frac{n^2}{r}\right) R(r) + \frac{d}{dr} R(r) + r\left(\frac{d^2}{dr^2} R(r)\right) = 0$$

Since the solution of this form of the equation will be

$> dsolve(q_{23}, R(r))$

$$R(r) = \_C1\ \mathrm{BesselJ}(n, \mu\, r) + \_C2\ \mathrm{BesselY}(n, \mu\, r)$$

we make the change of variables $r = \frac{x}{\mu}$ so that $R(r) = R\left(\frac{x}{\mu}\right) = y(x)$. This change is implement in Maple via

$> q_{24} := dchange\left(\left\{r = \frac{x}{\mu}, R(r) = y(x)\right\}, q_{23}, [y(x), x]\right)$

$$q_{24} := \left(x\,\mu - \frac{\mu\, n^2}{x}\right) y(x) + \mu\left(\frac{d}{dx} y(x)\right) + x\,\mu\left(\frac{d^2}{dx^2} y(x)\right) = 0$$

Under the assumption that $\mu \neq 0$, this equation simplifies to

$> q_{25} := collect\left(expand\left(\frac{q_{24}}{\mu}\right), y(x)\right)$

$$q_{25} := \left(x - \frac{n^2}{x}\right) y(x) + \frac{d}{dx} y(x) + x\left(\frac{d^2}{dx^2} y(x)\right) = 0$$

a Bessel equation of order n, and has general solution

$> dsolve(q_{25}, y(x))$

$$y(x) = \_C1\ \mathrm{BesselJ}(n, x) + \_C2\ \mathrm{BesselY}(n, x)$$

Homogeneous Dirichlet Condition

Continuity on $0 \le r \le c$ implies that $R(r) = \alpha\, J_n(\mu\, r)$, where α is constant. The homogeneous Dirichlet condition $u(c, \theta, z) = 0$ implies $R(c) = 0$, so that $\mu\, c$ must be a zero of $J_n(x)$. If we denote the kth zero of $J_n(x)$ by $BJZ(n, k)$, then $\mu_k^{(n)} = \frac{BJZ(n, k)}{c}$ represents a doubly-indexed set of eigenvalues for which the corresponding eigenfunctions are

$$F_k^{(n)}(r, \theta) = J_n\left(\frac{BJZ(n, k)}{c}\, r\right) \cos(n\theta),\ n = 0, 1, \ldots$$
$$G_k^{(n)}(r, \theta) = J_n\left(\frac{BJZ(n, k)}{c}\, r\right) \sin(n\theta),\ n = 1, 2, \ldots$$

To establish orthogonality of these eigenfunctions, we will have to evaluate the integral

$> q_{26} := \int_0^c r\,\mathrm{BesselJ}(n, \mu_j r)\,\mathrm{BesselJ}(n, \mu_k r)\,dr$

$$q_{26} := \int_0^c r\,\mathrm{BesselJ}(n, \mu_j r)\,\mathrm{BesselJ}(n, \mu_k r)\,dr$$

where the superscript (n) has been dropped from the distinct eigenvalues $\mu_j^{(n)}$ and $\mu_k^{(n)}$. For this integral Maple gives

$> \mathit{OrthogonalityRelation} := \mathit{value}(q_{26})$

$$\mathit{OrthogonalityRelation} := \frac{1}{\mu_j^2 - \mu_k^2}\left(c\left(\mu_k\,\mathrm{BesselJ}(n, \mu_j c)\,\mathrm{BesselJ}(n-1, \mu_k c) - \mu_j\,\mathrm{BesselJ}(n-1, \mu_j c)\,\mathrm{BesselJ}(n, \mu_k c)\right)\right)$$

which clearly vanishes by the definition of the eigenvalues $\mu_j^{(n)}$ and $\mu_k^{(n)}$.

The function $u(r, \theta, z)$ will be given by a Fourier-Bessel series in which a double sum is taken over both n and k. The coefficients in this series require us to evaluate the integral

$$\int_0^c r\,J_n^2\left(\mu_k^{(n)} r\right)\,dr$$

which we enter into Maple as

$> \mathit{NormRelation} := \int_0^c r\,\mathrm{BesselJ}(n, \mathrm{BesselJZeros}(n, m)\,r)^2\,dr$

$$\mathit{NormRelation} := \int_0^c r\,\mathrm{BesselJ}(n, \mathrm{BesselJZeros}(n, m)\,r)^2\,dr$$

We call this the "Norm Relation" since it is related to the L_2-norm of the eigenfunction.

Now it is indeed unfortunate that Maple incorrectly evaluates this integral to

$> \mathit{value}(\mathit{NormRelation})$

$$\frac{1}{\Gamma(2+n)\,\Gamma(n+1)}\left(2^{-2-2n} c^{1+2n} \sqrt{\pi}\,\mathrm{BesselJZeros}(n, m)^{-1+2n} (\mathrm{BesselJZeros}(n, m)\,c)^{-n}\,\mathrm{StruveH}(n, 2\,\mathrm{BesselJZeros}(n, m)\,c)\,\Gamma\left(n+\frac{3}{2}\right)\right)$$

an error that has been corrected for the next release of the product. The correct value of this integral is

$$\int_0^c r\, J_n^2\left(\mu_k^{(n)} r\right) dr = \frac{c^2}{2}\left[J_n'\left(\mu_k^{(n)} c\right)\right]^2$$

as can be found, for example, in the text Boundary Value Problems, Ladis D. Kovach, Addison-Wesley Publishing Company, 1984. Consequently, we have

$$\int_0^c\int_{-\pi}^{\pi} r\left[F_k^{(n)}(r,\theta)\right]^2 d\theta\, dr = \begin{cases} \frac{c^2}{2}\left[J_n'\left(\mu_k^{(n)} c\right)\right]^2 2\pi, & n = 0 \\ \frac{c^2}{2}\left[J_n'\left(\mu_k^{(n)} c\right)\right]^2 \pi, & n \geq 1 \end{cases}$$

$$\int_0^c\int_{-\pi}^{\pi} r\left[G_k^{(n)}(r,\theta)\right]^2 d\theta\, dr = \frac{c^2}{2}\left[J_n'\left(\mu_k^{(n)} c\right)\right]^2 \pi, \qquad n \geq 1$$

There is one additional subtlety to confront. Since $J_0(0) = 1$, the first eigenvalue in the sequence $\mu_k^{(0)}$ is not zero, so $k = 1, 2, \ldots$ corresponds to the indexing available via Maple's **BesselJZeros** command. Indeed, we see that

> $evalf(\text{BesselJZeros}(0, 1))$

2.404825558

is the first zero for J_0. However, $J_n(0) = 0$ for $n \geq 1$, so the "first zero" is $x = 0$, but this cannot be an eigenvalue because by definition, eigenvalues are numbers for which there are nontrivial solutions. Consequently, for the homogeneous Dirichlet condition, each sequence $\mu_k^{(n)}$ starts with $k = 1$ for any value of n. This indexing is consistent with Maple's **BesselJZeros** command where "1" in the second argument always gives the first *positive* zero.

Homogeneous Neumann Condition

The homogeneous Neumann condition $u_r(c, \theta, z) = 0$ implies $R'(c) = 0$, so that $\mu\, c$ must be a zero of $J_n'(x)$. If we denote the kth zero of $J_n'(x)$ by $\zeta(n, k)$, then $\mu_k^{(n)} = \frac{\zeta(n,k)}{c}$ represents a doubly-indexed set of eigenvalues for which the corresponding eigenfunctions are

$$F_k^{(n)}(r,\theta) = J_n\left(\frac{\zeta(n,k)}{c}\, r\right)\cos(n\theta),\ n = 0, 1, \ldots$$
$$G_k^{(n)}(r,\theta) = J_n\left(\frac{\zeta(n,k)}{c}\, r\right)\sin(n\theta),\ n = 1, 2, \ldots$$

To establish orthogonality of these eigenfunctions, we will have to show

> *OrthogonalityRelation*

$$\frac{c\left(\mu_k \operatorname{BesselJ}(n, \mu_j c)\operatorname{BesselJ}(n-1, \mu_k c) - \mu_j \operatorname{BesselJ}(n-1, \mu_j c)\operatorname{BesselJ}(n, \mu_k c)\right)}{\mu_j^2 - \mu_k^2}$$

vanishes. To this end, we seek to eliminate J_{n-1} and introduce J_n'. A generalization of the identity $J_0' = -J_1$ is

$> q_{27} := \frac{\partial}{\partial z}\operatorname{BesselJ}(n, z) = \frac{\partial}{\partial z}\operatorname{BesselJ}(n, z)$

$$q_{27} := \frac{\partial}{\partial z}\operatorname{BesselJ}(n, z) = -\operatorname{BesselJ}(n+1, z) + \frac{n \operatorname{BesselJ}(n, z)}{z}$$

Solving this for J_{n+1} leads to

$> q_{28} := isolate(q_{27}, \operatorname{BesselJ}(n+1, z))$

$$q_{28} := \operatorname{BesselJ}(n+1, z) = -\left(\frac{\partial}{\partial z}\operatorname{BesselJ}(n, z)\right) + \frac{n \operatorname{BesselJ}(n, z)}{z}$$

Replacing J_n' in this equation with its value from the previous equation gives

$> q_{29} := expand(eval(\operatorname{BesselJ}(n-1, z) = simplify(\operatorname{BesselJ}(n-1, z)), q_{28}))$

$$q_{29} := \operatorname{BesselJ}(n-1, z) = \frac{n \operatorname{BesselJ}(n, z)}{z} + \frac{\partial}{\partial z}\operatorname{BesselJ}(n, z)$$

Finally, substituting for J_{n-1} gives

$> expand\left(eval\left(OrthogonalityRelation, \left[q_{29}\Big|_{z=\mu_j c}, q_{29}\Big|_{z=\mu_k c}\right]\right)\right)$

$$\frac{c\, \mu_k \operatorname{BesselJ}(n, \mu_j c)\left(\frac{\partial}{\partial z}\operatorname{BesselJ}(n, z)\Big|_{z=\mu_k c}\right)}{\mu_j^2 - \mu_k^2} - \frac{c\, \mu_j \operatorname{BesselJ}(n, \mu_k c)\left(\frac{\partial}{\partial z}\operatorname{BesselJ}(n, z)\Big|_{z=\mu_j c}\right)}{\mu_j^2 - \mu_k^2}$$

from which it is evident that the functions $J_n\left(\mu_k^{(n)} r\right)$ are orthogonal, even when $\left\{\mu_k^{(n)} c\right\}$ are zeros of $J_n'(x)$.

Since $J_0'(0) = -J_1(0) = 0$, but $J_0(0) = 1$, then $\mu_0^{(0)} = 0$ is an eigenvalue and the indexing starts at $k = 0$. The corresponding eigenfunction is 1.

Maple computes $J_n'(0)$, $n \geq 1$, from the formula

> $\frac{\partial}{\partial x}\,\mathrm{BesselJ}(n,x) = \frac{\partial}{\partial x}\,\mathrm{BesselJ}(n,x)$

$$\frac{\partial}{\partial x}\,\mathrm{BesselJ}(n,x) = -\mathrm{BesselJ}(n+1,x) + \frac{n\,\mathrm{BesselJ}(n,x)}{x}$$

Consequently, a division-by-zero error occurs in, for example

> $\left.\frac{d}{dx}\,\mathrm{BesselJ}(1,x)\right|_{x=0}$

```
Error, (in VectorCalculus:-eval) numeric exception: division
by zero
```

However, an examination of the series expansions for $J_n'(x)$, $x \geq 1$, shows that this error is spurious. Indeed, we have

> **for** k **to** 5 **do**
  series$\left(\frac{\partial}{\partial x}\,\mathrm{BesselJ}(k,x),\, x=0\right)$
  end do;
  unassign('k')

$$\frac{1}{2} - \frac{3}{16}\,x^2 + \frac{5}{384}\,x^4 + \mathrm{O}(x^5)$$

$$\frac{1}{4}\,x - \frac{1}{24}\,x^3 + \mathrm{O}(x^5)$$

$$\frac{1}{16}\,x^2 - \frac{5}{768}\,x^4 + \mathrm{O}(x^5)$$

$$\frac{1}{96}\,x^3 + \mathrm{O}(x^5)$$

$$\frac{1}{768}\,x^4 + \mathrm{O}(x^5)$$

Each of these functions can be readily evaluated at $x=0$. Surprisingly, $J_n'(0)=0$ for $n \geq 2$, but $J_1'(0) = \frac{1}{2}$. Therefore, $x=0$ is not even a candidate for being an eigenvalue when $n=1$, and for greater values of n, although $x=0$ is a zero, $J_n(0)=0$ so that $x=0$ would not be an eigenvalue. Consequently, for the homogeneous Neumann condition, zero is an eigenvalue only for $n=0$.